

Part IV. Linear Differential Equations

Section 5. The Laplace Transform

In this section, examples show how to use *Mathematica* to obtain Laplace transforms and their inverses. The transform method for solving an ode and a linear system is illustrated. Applications to problems with discontinuous forcing functions are featured. See Ledder, Chapter 7.

Laplace Transforms and the Inverse Laplace Transform

We are interested in the *Mathematica* functions named **LaplaceTransform** and **InverseLaplaceTransform**. The syntax to generate the Laplace transform of an expression $f(t)$ as an expression $F(s)$ is

$$\mathbf{LaplaceTransform}[f(t), t, s]$$

The inverse Laplace transform of $F(s)$ to $f(t)$ is accomplished with

$$\mathbf{InverseLaplaceTransform}[F(s), s, t]$$

Four textbook examples:

1. The Laplace transform of $t e^{2t} \cos(3t)$

$$\mathbf{LaplaceTransform}[t \cdot \mathbf{Exp}[2 \cdot t] \cdot \mathbf{Cos}[3 \cdot t], t, s]$$

$$\frac{(-5 + s)(1 + s)}{(13 + (-4 + s)s)^2}$$

2. The inverse Laplace transform of $\frac{s^2-1}{s^3+8}$.

$$\mathbf{InverseLaplaceTransform}[(s^2-1)/(s^3+8), s, t]$$

$$\frac{1}{12} (3 e^{-2t} + 9 e^t \cos[\sqrt{3} t] - \sqrt{3} e^t \sin[\sqrt{3} t])$$

3. Add this to your Laplace transform table: The transform of $t e^{at} \sin(kt)$.

$$\mathbf{LaplaceTransform}[t \cdot \mathbf{Exp}[a \cdot t] \cdot \mathbf{Sin}[k \cdot t], t, s]$$

$$-\frac{2 \sqrt{k^2} (a - s) \text{Sign}[k]}{(k^2 + (a - s)^2)^2}$$

4. The transform of a discontinuous time function: $\cos(t) - t - \text{UnitStep}(t - 3)$

LaplaceTransform[Cos[t] - t - UnitStep[t-3],t,s]

$$-\frac{1}{s^2} - \frac{e^{-3s}}{s} + \frac{s}{1+s^2}$$

"UnitStep"? What is that all about.


Answer: The UnitStep function has its values defined as follows:

UnitStep[t] = 0 when t is negative

UnitStep[t] = 1 when t is positive.


See its graph below.

**Plot[UnitStep[t], {t,-2,2}, PlotRange->{-0.5,1.5},
PlotStyle->Thickness[0.015], AspectRatio->2/4]**



Consequently, UnitStep[t-3] is zero for t < 3 and 1 for t > 3. The UnitStep function and functions defined using it (like the one whose Laplace transform was found in 4) are called "piecewise continuous". The time function in Example 4 has a jump discontinuity at t = 3. Its graph looks like this.

Plot[Cos[t] + t - UnitStep[t-3], {t,-1,5}, PlotRange->{-1,3}]



Be on the lookout for UnitStep when inverting an s-domain function containing expressions of the form e^{-as} , a positive.


```
InverseLaplaceTransform[ s/(s^2+1) - 1/s^2 - Exp[-3*s]/s, s, t]
-t + Cos[t] - UnitStep[-3 + t]
```

And, be prepared to use UnitStep to deal with piecewise continuous functions. It happens all the time in engineering and science applications.

Typical driver: Off, then on, then off again

Suppose, for example, that an undamped mass spring system is set into motion from equilibrium with velocity 3 meters per second, then driven with a sinusoid for the 10 second interval between $t = 10$ and $t = 20$. See the graph below.

```
f[t_] := 20*(UnitStep[t-10] - UnitStep[t-20])*Sin[Pi*t/4]
Plot[ f[t], {t,0,40}, PlotRange->{-22,22}, AspectRatio->1/6,
PlotStyle->Thickness[0.005]]
```



Here is the Laplace transform of the driver.

```
LaplaceTransform[f[t], t, s]
20 ( (4 e^{-20 s} \pi) / (\pi^2 + 16 s^2) + (16 e^{-10 s} s) / (\pi^2 + 16 s^2) )
```

For the forced differential equation we will use $m = 40$ (kilograms), $b = 0$, $k = 10$.

```
DE = 40*y''[t] + 10*y[t] == f[t]
10 y[t] + 40 y''[t] == f[t]
```

The DSolve solution is messy.

```
soln = DSolve[ {DE, y[0]==0, y'[0]==3}, y[t], t ]
{{y[t] ->
  1 / (-4 + \pi^2) ( 2 ( -12 Sin[t/2] + 3 \pi^2 Sin[t/2] - 2 \pi Cos[t/2] Sin[10] UnitStep[-20 + t] +
    2 \pi Cos[10] Sin[t/2] UnitStep[-20 + t] + 4 Cos[t/2]^2 Sin[\pi t/4]
    UnitStep[-20 + t] + 4 Sin[t/2]^2 Sin[\pi t/4] UnitStep[-20 + t] + 4 Cos[5]
    Cos[t/2] UnitStep[-10 + t] + 4 Sin[5] Sin[t/2] UnitStep[-10 + t] - 4 Cos[t/2]^2
    Sin[\pi t/4] UnitStep[-10 + t] - 4 Sin[t/2]^2 Sin[\pi t/4] UnitStep[-10 + t] ) )}}
```

But it simplifies considerably.

soln = Simplify[soln]

$$\left\{ \left\{ y[t] \rightarrow \frac{1}{-4 + \pi^2} \left(6 (-4 + \pi^2) \sin\left[\frac{t}{2}\right] + \left(-4 \pi \sin\left[10 - \frac{t}{2}\right] + 8 \sin\left[\frac{\pi t}{4}\right]\right) \text{UnitStep}[-20 + t] + 8 \left(\cos\left[5 - \frac{t}{2}\right] - \sin\left[\frac{\pi t}{4}\right]\right) \text{UnitStep}[-10 + t] \right\} \right\}$$

The Laplace transform solution is much nicer looking. We can obtain it as follows.

Transform DE to the s-domain

LaplaceTransform[DE, t, s]

$$10 \text{LaplaceTransform}[y[t], t, s] + 40 (s^2 \text{LaplaceTransform}[y[t], t, s] - s y[0] - y'[0]) = 20 \left(\frac{4 e^{-20 s} \pi}{\pi^2 + 16 s^2} + \frac{16 e^{-10 s} s}{\pi^2 + 16 s^2} \right)$$

Substitute the initial values, and let Y replace LaplaceTransform[y[t],t,s]

%/.{y[0]->0,y'[0]->3,LaplaceTransform[y[t],t,s]->Y}

$$10 Y + 40 (-3 + s^2 Y) = 20 \left(\frac{4 e^{-20 s} \pi}{\pi^2 + 16 s^2} + \frac{16 e^{-10 s} s}{\pi^2 + 16 s^2} \right)$$

Solve for Y

Solve[%,Y]

$$\left\{ \left\{ Y \rightarrow \frac{4 e^{-20 s} (2 \pi + 3 e^{20 s} \pi^2 + 8 e^{10 s} s + 48 e^{20 s} s^2)}{(1 + 4 s^2) (\pi^2 + 16 s^2)} \right\} \right\}$$


Invert to get the solution formula (Compare to the simplified solution formula displayed above.)

InverseLaplaceTransform[Y/.,s,t]

$$\left\{ \frac{1}{-4 + \pi^2} \left(2 \left(3 (-4 + \pi^2) \sin\left[\frac{t}{2}\right] + \left(-2 \pi \sin\left[10 - \frac{t}{2}\right] + 4 \sin\left[\frac{\pi t}{4}\right]\right) \text{UnitStep}[-20 + t] + 4 \left(\cos\left[5 - \frac{t}{2}\right] - \sin\left[\frac{\pi t}{4}\right]\right) \text{UnitStep}[-10 + t] \right) \right\}$$

Here is a plot of the solution and the driver (between 10 and 20). The input graph has been scaled down to fit nicely into the plot. The driver causes the amplitude of the oscillations to decrease from 6 to 4 meters (approximately). The phase of the oscillations has also been affected, but not the period.

```
Plot[ {y[t]/.soln,f[t]/7}, {t,0,40}, AspectRatio→1/3]
```



The Dirac delta: DiracDelta

The Dirac delta function is written as **DiracDelta** in *Mathematica*.

```
LaplaceTransform[DiracDelta[t-3], t, s]
```

$$e^{-3s}$$

```
InverseLaplaceTransform[1, s, t]
```

```
DiracDelta[t]
```

Lets drive the mass spring system introduced above with a blow of magnitude 200 in the positive direction at time $t = 12$. Note that DE[[1]] can be used to refer to the left side the DE equation.

```
DE2 = DE[[1]] == 200*DiracDelta[t-12]
```

```
10 y[t] + 40 y''[t] == 200 DiracDelta[-12 + t]
```


The solution formula, using the same initial conditions, looks like this.

```
soln = DSolve[ {DE2, y[0]==0, y'[0]==3}, y[t], t]
```

```
{ {y[t] → 2 ( 3 Sin[ t/2 ] -
5 Cos[ t/2 ] Sin[6] UnitStep[-12 + t] + 5 Cos[6] Sin[ t/2 ] UnitStep[-12 + t] ) } }
```

The following plot displays the solution curve (output) and vertical line placed at the time of the application of the blow (input).


```
Show[ Plot[ y[t]/.soln, {t,0,40} ],
      ListPlot[ {{12,0},{12,5}}, PlotJoined->True,
      PlotStyle->Thickness[0.015] ], AspectRatio->1/3 ]
```



The effect of the blow is to cause a kink in the curve for position as a function of time. There will also be a jump discontinuity on the curve for velocity.

The phase plane plot below shows the discontinuity in velocity as a vertical line joining the initial elliptical trajectory to the elliptical trajectory describing the motion of the system after the blow.

```
Trajectory = {y[t]/.soln[[1]],D[y[t]/.soln[[1]],t]};
ParametricPlot[ Trajectory, {t,0,22}, AxesOrigin->{-20,-10},
      PlotRange->{{-20,20},{-10,10}},
      AxesLabel->{"POSITION", "VELOCITY"}]
```



Applying the Laplace transform method to a linear system

The Laplace transform method can also be applied to a system of linear equations (constant coefficients), driven or undriven. However, if the system is driven by discontinuous forcing functions, the Laplace transform solution formulas may contain complex exponentials. Consider the following IVP

$$x' = x - y + 10 \text{ UnitStep}(t - 2)$$

$$y' = x + y$$

$$x(0) = 1, y(0) = 1$$

```

DEsystem = {x'[t] == x[t] - y[t] + 10*UnitStep[t-2],
          y'[t] == x[t] + y[t] }
{x'[t] == 10 UnitStep[-2 + t] + x[t] - y[t], y'[t] == x[t] + y[t]}

```

Here is the DSolve solution for the initial conditions $x(0) = 1$ and $y(0) = 1$.

```

DSsoln = DSolve[ Join[DEsystem,{x[0]==1, y[0]==1}], {x[t],y[t]}, t]
{{x[t] ->  $\frac{1}{e^2} (e^{2+t} \cos[t] - e^{2+t} \sin[t] +$ 
          5 et Cos[2] Cos[t] UnitStep[-2 + t] - 5 e2 Cos[t]2 UnitStep[-2 + t] -
          5 et Cos[t] Sin[2] UnitStep[-2 + t] + 5 et Cos[2] Sin[t] UnitStep[-2 + t] +
          5 et Sin[2] Sin[t] UnitStep[-2 + t] - 5 e2 Sin[t]2 UnitStep[-2 + t]),
          y[t] ->  $\frac{1}{e^2} (e^{2+t} \cos[t] + e^{2+t} \sin[t] - 5 e^t \cos[2] \cos[t] \text{UnitStep}[-2 + t] +$ 
          5 e2 Cos[t]2 UnitStep[-2 + t] - 5 et Cos[t] Sin[2] UnitStep[-2 + t] +
          5 et Cos[2] Sin[t] UnitStep[-2 + t] -
          5 et Sin[2] Sin[t] UnitStep[-2 + t] + 5 e2 Sin[t]2 UnitStep[-2 + t])}}

```

Is simplifies as follows.

```

DSsoln = Simplify[{x[t],y[t]}/.DSsoln]
{{  $\frac{e^{2+t} (\cos[t] - \sin[t]) - 5 (e^2 - e^t \cos[2 - t] + e^t \sin[2 - t]) \text{UnitStep}[-2 + t]}{e^2},$ 
           $\frac{e^{2+t} (\cos[t] + \sin[t]) - 5 (-e^2 + e^t \cos[2 - t] + e^t \sin[2 - t]) \text{UnitStep}[-2 + t]}{e^2}$ }}

```

The Laplace transform solution will arrive in complex form. See below.

Transform the system

```

LaplaceTransform[DEsystem, t, s ]
{s LaplaceTransform[x[t], t, s] - x[0] ==
           $\frac{10 e^{-2 s}}{s} + \text{LaplaceTransform}[x[t], t, s] - \text{LaplaceTransform}[y[t], t, s],$ 
          s LaplaceTransform[y[t], t, s] - y[0] ==
          LaplaceTransform[x[t], t, s] + LaplaceTransform[y[t], t, s]}

```

Substitute the initial values and X and Y for the transforms of the solution

```

%/.{x[0]->1,y[0]->1,LaplaceTransform[x[t],t,s]->X,LaplaceTransform[y[t],t,s]-
>Y}
{-1 + s X ==  $\frac{10 e^{-2 s}}{s} + X - Y, -1 + s Y == X + Y}$ 

```

Solve for X and Y

Tsoln = Solve[%, {X, Y}]

$$\left\{ \left\{ X \rightarrow -1 + \frac{(1-s) \left(\frac{10 e^{-2s}}{s} + s \right)}{-1 - (-1+s)^2}, Y \rightarrow -\frac{\frac{10 e^{-2s}}{s} + s}{-1 - (-1+s)^2} \right\} \right\}$$

Invert to obtain the solution trajectory (Note that the solution is in terms of complex exponentials)

Lsoln = InverseLaplaceTransform[{X, Y]/.%, s, t]

$$\left\{ \left\{ \left(\frac{1}{2} + \frac{i}{2} \right) e^{-i((2-2i)+t)} \left(e^{(2+2i)+t} (-i + e^{2it}) + 5((-1+i) e^{(2+2i)+it} - i e^{(1+2i)t} + e^{4i+t}) \text{UnitStep}[-2+t] \right), \left(\frac{1}{2} - \frac{i}{2} \right) e^{(1-i)t} (i + e^{2it}) - \left(\frac{5}{2} + \frac{5i}{2} \right) ((-1+i) - i e^{(1-i)(-2+t)} + e^{(1+i)(-2+t)}) \text{UnitStep}[-2+t] \right\} \right\}$$

The function called **ComplexExpand** will change the complex exponentials to Cartesian form, but the output is very long. We suppress it, then **Simplify**. Compare the output to the simplified DSolve solution shown above.

ComplexExpand[Lsoln];

Simplify[%]

$$\left\{ \left\{ \frac{e^{2+t} (\text{Cos}[t] - \text{Sin}[t]) - 5 (e^2 - e^t \text{Cos}[2-t] + e^t \text{Sin}[2-t]) \text{UnitStep}[-2+t]}{e^2}, \frac{e^{2+t} (\text{Cos}[t] + \text{Sin}[t]) - 5 (-e^2 + e^t \text{Cos}[2-t] + e^t \text{Sin}[2-t]) \text{UnitStep}[-2+t]}{e^2} \right\} \right\}$$

Plot of the solution trajectory

The solution trajectory is plotted below along with a point plotted for every quarter unit of time. See if you can identify the point where the UnitStep driver starts to have an effect on the output.


```
Show[ ListPlot[Table[DSsoln[[1]],{t,0,4,1/4}],PlotStyle->PointSize[0.02]],  
ParametricPlot[ DSsoln, {t,0,4}, PlotStyle->RGBColor[0,1,0]],  
AspectRatio->1/1, PlotRange->{{-15,25},{-25,5}},  
AxesOrigin->{-15,-25}, AxesLabel->{"x","y"} ]
```

